

Walker:

Working with a Multi-button Keypad in GtkAda

Rick Duley

Perth, Western Australia
2009

Table of Contents

1	Preamble	1
2	Introduction	1
2.1	Glade	1
2.2	Why Verbose Code?	2
2.3	Naming Parameters.....	2
2.4	Initializing Scalar Variables	2
2.5	Connotative Identifiers.....	2
2.6	On Being Methodical.....	3
2.6.1	The Plan	3
2.6.2	Development Folder Structure.....	3
3	Getting GtkAda.....	3
3.1	Setting Up GAP's GtkAda.....	3
3.2	GtkAda Documentation.....	3
4	Preparations	3
4.1	The Mud Map	3
5	Create the Key	4
5.1	General Issues with Gtk_Key_Button.....	4
5.1.1	Extending Gtk_Button	4
5.1.2	Privacy	4
5.1.3	Using Access as a Parameter	5
5.2	Implementing Key_Button	5
5.2.1	Converting Integer to Character	5
5.2.2	Naming the Key_Button	5
5.2.3	Creating the Button Label	5
5.2.4	Including Optional Icons and Labels.....	5
5.3	Testing Gtk_Key_Button.....	6
5.3.1	Creating an Icon Image.....	6
5.3.2	Sneak a Peek.....	6
6	Create the Keypad.....	6
6.1	Keypad Issues	6
6.1.1	Boxes for Widgets	6
6.1.2	Using Constants.....	6
6.1.3	Named Numbers	7
6.1.3.1	Advantages of Named Numbers	7
6.1.4	Keys_Type.....	7
6.1.5	Privacy	7
6.1.6	Handling Callbacks for the Key_Buttons	7
6.2	Keypad Implementation.....	7



6.2.1	Storing Data for Individual Buttons.....	7
6.2.2	Sneak a Peek.....	8
7	Walker.....	8
7.1	Button Callbacks from the Keypad.....	8
7.1.1.1	The Appropriate Callback Package.....	8
7.1.1.2	The Appropriate Callback Procedures.....	9
7.1.1.3	Calling the Appropriate Procedures.....	9
7.2	Making Walker Walk.....	9
8	Finished.....	10
9	Source Code.....	11
9.1	Key_Button.....	11
9.1.1	Key_Button Specification.....	11
9.1.2	Key_Button Body.....	13
9.1.3	Key_Button_Test.....	16
9.2	Keypad.....	17
9.2.1	Walker_Keypad Specification.....	17
9.2.2	Walker_Keypad Body.....	19
9.2.3	Walker_Keypad_Test.....	23
9.3	Walker.....	23
9.3.1	Walker_Pkg Specification.....	23
9.3.2	Walker_Pkg Body.....	25
9.3.3	Walker_Pkg_Callback_Pkg Specification.....	28
9.3.4	Walker_Pkg.Callbacks Specification.....	28
9.3.5	Walker_Pkg.Callbacks Body.....	28
9.3.6	Walker_Keypad_Pkg.Local_Callbacks Specification.....	29
9.3.7	Walker_Keypad_Pkg.Local_Callbacks Body.....	30
9.3.8	Walker_Keypad_Operations Specification.....	31
9.3.9	Walker_Keypad_Operations Body.....	33
9.3.10	Walker — Mainline.....	35
10	Access Types as Parameters.....	36
11	Bibliography.....	37



Walker: Working with a Multi-button Keypad in GtkAda

Rick Duley
Perth, Western Australia

Under Australian Law (The Copyright Act of 1968 and Amendments) this work is copyright to the author by fact of its creation and commission to a storage medium. Proprietary rights are hereby waived provided the work is used for educational purposes, AND that the work is presented in full and unmodified, AND that this copyright remains attached and unmodified, All moral rights under the Australian Copyright Amendment (Moral Rights) Act of 2000 are reserved.

1 Preamble

The author would like to acknowledge the gallant and patient assistance of the people from AdaCore, the Gnat Academic Program, the GtkAda Mailing List, CompLangAda newsgroup, and others without which he would not have made it this far. This document is humbly presented in the hope that its availability will lessen those people's future torment. Thanks folks!

2 Introduction

Many, maybe most, Graphical User Interfaces (GUIs) use buttons. They appear on their own, they appear in rows, sometimes they appear in keyboards. They have text labels, they have icons. They are one of the usual means of GUI control of an application.

This exercise is actually much more than an exercise with `Gtk.Button`. In this exercise you will create in verbose Ada code a keyboard of four buttons, set them with icons instead of labels and make them move the GUI around in the screen.

Although the GUI in this application is considerably simpler than the GUI in "*Gadget: Building a Compound Widget with GtkAda*", the interesting material in this exercise is in making the application do something — and in how we make it do something. This exercise

deals with the use of constants, arrays of widgets, arrays of `Unbounded_String`, access to procedure types, arrays of access to procedure, and a list of other things.

For resolution of matters not dealt with here go to <http://libre.adacore.com/libre/tools/gtkada/> and register with the GtkAda Mailing list. (Go to the bottom of the page and the paragraph beginning, "*Note that the ...*".

2.1 Glade

GtkAda from AdaCore comes with an automated developer called Glade. This lets a programmer create a Graphical User Interface interactively, then it creates the code to suit. I unashamedly used this package to learn many of the rudiments of GtkAda programming.

However, the true power and range of the GtkAda package is only realised by manually coding it, and learning some of the skills required to do that is the object of this exercise. One of those skills is to write code so that maintenance people can understand it. Reading machine-written code in general, and Glade's is no exception, is like trying to divine something by gazing at the entrails of a hen.

You can read about Glade and the way it works in the GtkAda User Guide under the heading "*Support for Glade, the Gtk GUI builder*".



2.2 Why Verbose Code?

One of the problems with newcomers learning GtkAda by working with Glade is that Glade code does not specify the origin of the widgets and things it uses. Glade religiously implements the ‘use’ clause. Consequently, beginners do not learn to work with the GtkAda Reference Manual. Because of Glade they get code that works, but they do not get to know how it works.

Verbose style programming changes this because the relevant package is named at each procedure or function call. Furthermore, many of the procedures and functions implemented are inherited from ancestor widgets. Programmers must learn to trace the ancestry of the current widget to find the appropriate operation.

Verbose style programming means a lot of typing, but, in the end, it is quicker for a beginner (and miles better for the maintainer). Programmers tend to think more carefully about each step. Therefore, they avoid unending confusion. Time spent typing is more than compensated for with time saved debugging.

2.3 Naming Parameters

Meticulously naming formal parameters and assigning values to them forces programmer to think carefully about those values as well. GAP’s GPS 2009 (see paragraph 3) is a great help here because it offers the parameters of a function or procedure call as you write. That certainly saves having to look up the Reference Manual for a parameter list every five minutes.

Another error, which is frequent when assigning parameter values by association, is that of getting the parameters in the wrong order. This glitch can be one of the hardest to find. This virtually never happens in verbose programming, and if it does parameter order doesn’t matter as long as the Actuals match the Formalis.

2.4 Initializing Scalar Variables

If you are not in the habit of initializing variables, try a test program like Code Sample 1. Some compilers are sensitive to uninitialized variables and this can cause problem when attempting to produce a ‘Release’ version of the executable.

```
with Ada.Text_IO;
use Ada.Text_IO;
with Ada.Integer_Text_IO;
use Ada.Integer_Text_IO;
procedure Test_Uninitialized_Variables is
  Int : Integer;
  Char : Character;
begin
  Put(':');
  Put(Int);
  Put(':');
  New_Line;
  Put(': ' & Char & ':');
end;
```

Code Sample 1: Test Uninitialized Variables

Note well: it pays to initialize variables. In this exercise initialization is carried to an extreme — access types are automatically initialized to ‘null’ — just to make the point.

2.5 Connotative Identifiers

Make sure the names of the widgets are distinctive and self-explanatory at the time you draw your Mud Map (see page 3), and use them when you program. Call the widgets anything you wish, but make sure the name describes what the widget or variable does. Do not accept the identifiers I used as any ultimate.

This idea of the use of connotative identifiers is one of the foundation principles of Ada Programming — which is what you will be doing in this exercise. The Style Manual still offers these suggestions:

- choose names that are as self-documenting as possible;
- use names given by the application;
- do not use obscure jargon;
- avoid using the same name to declare different kinds of identifiers;
- do not use an abbreviation of a long word when a shorter synonym exists;
- use a consistent abbreviation strategy;
- do not use ambiguous abbreviations;
- to justify its use, an abbreviation must save many characters over the full word;
- use abbreviations that are well-accepted in the application domain; and
- maintain a list of accepted abbreviations, and use only abbreviations on that list. (Software Productivity Consortium, 1995)



2.6 On Being Methodical

All programming is methodical in nature — graphics programming especially so. Consider the exploded construction diagram on page 8. ***Don't Panic!*** For the moment, note only that it starts with the base widget (the window itself) and progressively, working to the right, adds further widgets. In turn, each of these widgets is created, its characteristics are defined, and it is added to the widget to its left. (There are other things that go on but they are not important just yet.) Keep this idea of being methodical in mind. It is the basis of the whole process.

As St Paul put it,

"Let all things be done decently and in order."— 1 Corinthians 14:40.

2.6.1 The Plan

1. Create a button that will know where it is in a keypad;
2. Create a keypad of these buttons with each button aware of its position;
3. Use the keypad in an application widget.

2.6.2 Development Folder Structure

This methodical approach is reflected in the structure of the development folders during application development under GAP's GPS 2009. There is one folder where the key_button is developed and tested, another where the keypad is developed and tested, and a third for developing and testing the application. Code in these folders is interlinked by the GPS program files (.GPR). Walker.gpr is headed by "with ../../Walker_Keypad/Walker_Keypad.gpr"; and Walker_Keypad.gpr is headed by with ../../key_button/Key_Button.gpr";. Each project file calls its predecessor. This permits completely individual testing of components.

Why do they all have image subfolders? Because the images — at least some of them — are needed at each development stage.

When using other systems, work in the one folder.

3 Getting GtkAda

For a great start contact gap-contact@adacore.com and get registered




Figure 1: Development Folder Structure

with the GNAT Academic Program run by AdaCore. Then you can download the GAP Package which includes a quality Integrated Development Environment, complete with compiler and support packages including Win32 and GtkAda. The best advice is to download and install the lot using the default installation settings.

3.1 Setting Up GAP's GtkAda

Some modifications have to be made to GAP's GtkAda installation for PC. Full details are available at AdaSafeHouse.webs.com/GtkAda.html.

3.2 GtkAda Documentation

Full HTML documentation is included in the GtkAda package. If you have used all the default settings it can be found at `c:\GtkAda\share\doc\gtkada\`. I refer to these documents all the time so I have shortcuts to them on my desktop. Information on the functions of widgets and the meanings of formal parameters presented here is largely taken from these documents.

4 Preparations

Assuming that you have worked through *"Gadget: Building a Compound Widget with GtkAda"* you will know that building such a widget is like building something with LEGO blocks. You simply take the pieces and plug them together. However, the programmer must know exactly what is to be built before coding starts. Anything other than forethought and planning in fine detail is a recipe for frustration and futility. **Be warned!**

4.1 The Mud Map

Programmers should draw what Australians call a *Mud Map* before doing



anything else. A Mud Map, strictly speaking, is a sketch scraped in the dirt with a finger or a stick. It's just something to get a rough idea. Draw that sort of a sketch of the layout to be achieved including all the visible and invisible widgets to be included in the compound widget. Leave nothing out, and name all the parts in a logical, distinctive and connotative manner. Remember, an exploded view of Walker is shown on page 8 and that may make Figure 2 more clear.

5 Create the Key

If you have worked through “*Gadget: Building a compound widget with GtkAda*” you will have few problems with the code in `Gtk.Key_Button_Pkg` starting on page 11. A `Key_Button` is merely an extension of `Gtk.Button`.

5.1 General Issues with `Gtk_Key_Button`

There are, however, some issues in the implementation of `Gtk_Key_Button` to which we must pay some attention.

5.1.1 Extending `Gtk_Button`

Keys in a keypad must carry information as to where they are in the pad. (It is up to the programmer to know what each key is supposed to do.)

There are two obvious ways of doing this: (1) keep a record of the row and column at which the button appears;

Both are employed for `Key_Button`. In fact, only the Row and Column numbers are used in this project, but a Software Engineer would utilize both sets so that the same `Key_Button` can be used in other projects in which the Name might be used.

5.1.2 Privacy

Ada Packages provide ‘*Module Encapsulation*’ (see Barnes, 2006, p.42) by which the inner working of a package, as implemented in the package body, are not visible to the user. Declarations in the specification of a package are visible to the user, and can, therefore, be modified by the user. This is not necessarily a good thing from a Software Engineering point of view. In Walker, such access is definitely not desirable.

One thing about keeping the Row and Column numbers — they must stay unchanged when set. Imagine what would happen if keys in the pad behaved as if they were other keys! This means the developer does not want the user to change the values. At the same time, the differences between `Key_Button_Record` and ordinary `Gtk.Button`. `Gtk_Button_Record` must be known to the compiler. Ada’s solution is to declare the actual extension of the record to be ‘**private**’ (see line 9 of the `Gtk.Key_Button_Pkg` specification on page 11).

*The part of the package specification before the reserved word **private** is*

4




Figure 2: Walker Mud Map

(2) name the button for its position. | *the visible part and gives the*

DOCUMENT CURRENT ONLY WHEN PRINTED



information available externally to the package. ... [The details of the extension are declared after the word **'private'**.] ... This means that outside the package nothing is known of the details of the type. (Barnes, 2006, p. 221)

In other words, outside `Gtk.Key_Button_Pkg` or a child of `Gtk.Key_Button_Pkg` the record variables `Row` and `Column` cannot be addressed¹. Package authors can provide for this by writing procedures and functions which will access the record entities. For the use of developers, the functions `Get_Button_Row` and `Get_Button_Column` are implemented at lines 87 and 94 of the body of `Gtk.Key_Button` (starting on page 13) and used in lines 32 and 33 of `Walker_Keypad_Pkg.Local_Callbacks.adb` on page 31. Don't worry too much about this yet.

While they may not be directly addressed from outside the package, the inner details of the widget can be read by developers using the package which gives them an understanding of those details. In `Gtk_Key_Button`, `Label`, `Image`, `Box`, `Row` and `Column` for `Gtk_Key_Button_Record` are defined in lines 51-55 in the specification of `Gtk.Key_Button`.

5.1.3 Using Access as a Parameter

Declaring the parameters to `Get_Button_Row` and `Get_Button_Column` as:
access `Gtk_Key_Button_Record`'Class may seem a longwinded way of declaring them as `Gtk_Key_Button`, but it is necessary. See the discussion on the Google Group 'CompLangAda' starting on page 36. `CompLangAda` (<http://groups.google.com/group/comp.lang.ada/topics>) brings a variety of very wise heads together, and opinions vary. It pays to follow the thread for a while — and it pays to read right through this subset of responses to my request for an explanation — if you want to get to the bottom of a matter.

¹ Inside the package private entities can only be accessed for assignment or equality.

5.2 Implementing Key_Button

Here we work through the body of `Gtk.Key_Button` looking at the implementation issues.

5.2.1 Converting Integer to Character

Lines 36 and 36 in `Gtk.Key_Button.adb` starting on page 13 show how to convert an Integer into the corresponding character. This technique is an outcome of the definition of the standard type `Character` in the Ada Language Reference Manual (LRM) Section A.1.

The alphanumeric characters start at the character '0' (nought) with the position 48. So we use that offset of the position of the individual numeral to get the character position:

1. Let `Row = 3`
2. `3 + 48 = 51`
3. The Character at position 51 in LRM A.1(35/2) is '3'

At line 46 in the body of `Gtk.Key_Button` that actual character is concatenated into the `Name` of the button.

5.2.2 Naming the Key_Button

For `Gtk_Key_Button` the naming convention is `Row & '_' & Column`. Therefore, if the button is at `Row=1` and `Column=1` its name is to be "1_1". This algorithm is implemented at line 45 in the body of `Gtk.Key_Button`.

5.2.3 Creating the Button Label

When the button is initialized at line 41 in `Gtk.Key_Button.adb`, the parameter `'Label'` is left at its default value of a null string. At initialization, then, the button has no label. We want to create one and that is done at line 70; that is to say, it is done if `'Initialize'` receives a non-null UTF8_String as the actual for `"Label"`.

5.2.4 Including Optional Icons and Labels

It may be that the user of the package `Gtk.Key_Button` does not want an icon on the `key_button` (e.g. a key on a computer keyboard); it may be that the user wants only an icon (e.g. a 'STOP' sign on a stop button). The Software Engineer will allow for both. This is





Figure 3: Key_Button_Test

achieved by the conditional statements at lines 58 and 68.

If the user has not included an actual `Gtk.Image` in the parameters for the call to `Gtk_New` (i.e. `Image = null`), no image is added to 'Box'. In the same way, if no label text is specified in the parameters, no label is added to 'Box'. If neither is specified you take your chances with the shape of the button.

5.3 Testing Gtk_Key_Button

So, you have built a successful `Gtk.Key_Button`. How do you know?

It is always good practice to test a new package of anything in Ada when you have completed it — and, for larger packages, at stages through its development.

Note well: it is always easier to find bugs when the amount of code you have to search in is small. Frequent, incremental, testing is the salvation of many hair follicles.

5.3.1 Creating an Icon Image

There are several different versions of `Gtk.Image`.`Gtk_New` (see the `GtkAda` Reference Manual). The one used in line 19 of `Gtk.Key_Button.adb` will accept BMP, JPG, GIF, and PNG image files. In `Gtk_Key_Button` the images were GIF to take advantage of the transparency facility which shows the image without a background.

5.3.2 Sneak a Peek

Write a test program like the one starting on page 16. If your code is correct you should see something like Figure 3.

6 Create the Keypad

Walker's keypad has just four buttons. The trick here is to get the buttons into a `Gtk_Table` (Keypad) in a way in which we can address them. Our solution is to use three arrays: one for the images (the arrow symbols), one for the boxes

holding the images and the third for the Table (Keypad) which presents the buttons neatly set out in the screen display. More of that as we go.

6.1 Keypad Issues

Some more urgent general points to consider:

6.1.1 Boxes for Widgets

It is a safe practice to place each widget separately in its own `HBox` or `VBox`. There are two related issues to deal with here:

1. The `Gtk_Table` (Keypad) is, to all intents and purposes, another kind of box — both `Gtk.Table` and `Gtk.Box` are direct descendants of `Gtk_Container`. It is simply a special box in which widgets are displayed in rows (down) and columns (across). Therefore, we do not need a box for the button.
2. Keypad forms the base of a distinct widget — `Walker_Keypad`. Why then use `Keypad_Box` to hold the keypad? It is certainly not necessary because Keypad's outer layer is, itself, a box. Simply, it is good practice because the programmer may not be aware of the nature of the outer layer of a widget. Do you know the nature of the outer layer of any of the `GtkAda` standard widgets?

6

6.1.2 Using Constants

It is not bad practice to use Global Constants from a package specification. In contrast, variables declared in a specification should be passed as parameters and not used Globally (Dale et al., 2000, p. 314).

Constants are replaced in the program by the value represented, so there is no need to look up that value each time as must be done for a Global Variable; hence there is an execution speed advantage. However, the greatest advantage in using them this way, rather than using Literal values, is that there is only one place where the programmer needs to change them should the need arise.

For example, if Walker needed an extra row of keys only `KEYPAD_ROWS_COUNT` (see line 13 in `Walker_Keypad_Pkg.ads` starting on



page 17) would have to be changed². That is a whole lot simpler than going through the entire program changing Literals, and also stops the generation of very strange and frustratingly obscure errors. Some errors like that are just funny, others cause programs to crash for no apparent reason. Avoid the hassles, **make use of Constants**.

6.1.3 Named Numbers

You will note that the two constants declared at lines 13 and 16 in Walker_Keypad_Pkg.ads do not have specified types. They have become what is known as “Named Numbers”.

It is usually good practice to omit the type when declaring numeric constants ... But note that the type cannot be omitted in numeric variable declarations even when an initial value is provided.” (Barnes, 2006, p. 75)

Integer numbers declared in a number declaration ... such as

```
Ten : constant := 10;
```

are also of type universal integer. However, there are no universal integer variables and no universal integer operations. ... since these expressions are all static and thus evaluated at compile time the distinction [from constants] is somewhat theoretical ... However, it is an important rule that a static expression may not exceed the base range of the expected type. (Barnes, 2006, p. 375)

6.1.3.1 Advantages of Named Numbers

In lines 58 and 59 of Walker_Keypad_Pkg.adb the constants KEYPAD_ROWS_COUNT and KEYPAD_COLUMNS_COUNT are sent as parameters required to be of the type ‘Glib.Gint’. No type conversion is necessary as in lines 88 to 91 where Positive variable values have to be coerced into Gint. By not specifying the type KEYPAD_ROWS_COUNT can be used as any whole number type providing it fits in the range of that type.

² The compiler would also force the programmer to add more cells to some arrays and make other related changes.

6.1.4 Keys Type

(See line 19 in Walker_Keypad_Pkg.ads)

There is no mechanism to address widget in a Table directly. Instead, the keypad has an array of key_buttons (Pad) which can be addressed directly. These are linked to Keypad for display. The link is by means of an access type, so the changes made to Pad will show, where applicable, in Keypad. In fact, the changes made from outside Walker_Keypad_Pkg are non-visible, the connection of callbacks. More of that later.

6.1.5 Privacy

In line 24 of Walker_Keypad_Pkg.ads the extension to Gtk_Table_Record is declared to be **private**. (See discussion in paragraph 5.1.2 on page 4.) Buttons in ‘Pad’ need to be accessible for assignment when the user attaches callbacks to the buttons so a function Get_Walker_Keypad_Button is provided in line 44 of Walker_Keypad_Pkg.ads (for its use see line 68 of Walker_Pkg.adb).

6.1.6 Handling Callbacks for the Key Buttons

Key_Button_Handler_Pkg is declared at line 37 in Walker_Keypad_Pkg.ads. The purpose of this package (which implements a variation of Gtk.Handlers.User_Callback) is to deal with a click on one of the buttons. It is declared locally so it has visibility of the Keypad.

6.2 Keypad Implementation

Initialization of the Walker_Keypad is simplified by use of a loop structure.

6.2.1 Storing Data for Individual Buttons

Perhaps the advantage over individual creation and assignment of the buttons in the pad is not great with a keypad of four buttons. Spare a thought, however, for the creator of a keypad of 100 buttons! This loop operation, however, requires a new data structure to hold data relevant to the individual buttons. This is Button_Arrow_Data_Type at line 11 in Walker_Keypad_Pkg.adb

Constants of this type are referred to in lines 68 and 77 where they give the





Figure 4: Walker_Keypad_Test

address of the arrow image and the label text for the button respectively. Two things to note:

1. Being declared in the body of the package, this type and these constants are **extremely private**;
2. This is another example of storing information in one place so that if it needs modification it only needs to be modified in one place.

6.2.2 Sneak a Peek

Write a test program like the one starting on page 23 and you should have something that looks like Figure 4. It still doesn't do anything, that is the next item in the plan.

7 Walker

Until now, this project has, with a few exceptions, been a repetition of the 'Gadget' project, just playing with GtkAda widgets. This section is Ada rather than GtkAda and we deal with things like access types that access subroutines, and arrays of those types. We also dabble in the concept of inheritance as we set callbacks for the buttons on our keypad.

The callback to the window destructor (connected at line 35 of Walker_Pkg.adb) is normal and is handled at Walker_Pkg level. The relevant handler package is in Walker_Pkg_Callback_Pkg (see page 28) and the handler procedure (On_X_Button_Destroy) in Walker_Pkg.Callbacks starting on page 28. However, the callbacks from the buttons present a visibility problem.

7.1 Button Callbacks from the Keypad

Handling callbacks from the buttons in the keypad requires a callback package




Figure 5: Walker - Exploded View






Figure 6: From Array Parameter to Procedure

which is a child of the keypad package. This is Walker_Keypad_Pkg.Local_Callbacks (see page 29) and the connection of the callback procedure at line 69 of Walker_Pkg.adb should be paid careful attention.

7.1.1.1 The Appropriate Callback Package

To handle messages from the keys in the keypad, this connection uses Key_Button_Handler_Pkg which is declared at line 37 of Walker_Keypad_Pkg.ads on page 18. This, remember, is a special version of Gtk.Handlers.User_Callback and it requires access to the keypad.

7.1.1.2 The Appropriate Callback Procedures

The callback procedures (see On_Walker_Keypad_Button_Click in Walker_Keypad_Pkg.Local_Callbacks on page 29) also need some special information:

1. The callback needs to know the coordinates at which the window is displayed on the screen. These are collected at lines 16 and 21 of Walker_Keypad_Pkg.Local_Callbacks;
2. It also needs the 'handle' of the window (so that it can destroy the window and all it contains). This is collected at line 25.

7.1.1.3 Calling the Appropriate Procedures

Walker_Pkg_Operations.ads (see page 31) holds the key to how signals from the keys in the keypad are converted into actions on the displayed widget. The secret is the 'access to subprogram' type described at LRM 3.10 (6/2). At line 13 in Walker_Pkg_Operations we declare the array type Operation_Access_Array_Type (which is a dimensional match for keypad). At line 35 we have the constant of that type Operation_Access_Array which serves as index into a list of available operations. In turn, this index is searched by the procedure call in line 32 of Walker_Keypad_Pkg.Local_Callbacks on page 31. Here the coordinates of the array are collected by calls to the functions Get_Button_Row and Get_Button_Column in Gtk.Key_Button and are relative to the button's position on the keypad. It sounds complicated, but, when you understand what is going on, it is a simple solution (especially for keypads with large numbers of keys). Figure 6 may clarify the issue.

7.2 Making Walker Walk

By default, window positions are defined by the top-left corner of the window. The coordinate system used for a screen in windows starts at (0, 0) in the top-left of the screen. The range used in Walker is:

- X = 100 → left;



- X = 400 → right;
- Y = 100 → top;
- Y = 400 → bottom.

So, wherever Walker might be on the screen, the 'Go Up' key is to cause the window's Y-coordinate to be 100. The X-coordinate will remain unchanged. (See line 39 in Walker_Pkg_Operations on page 34.)



Figure 7: Walker - Complete

8 Finished

The mainline program is simple (see page 35). Walker should look something like Figure 7 and the window should move about the screen when the buttons are clicked.



9 Source Code

9.1 Key_Button

9.1.1 Key_Button Specification

```
1 with Gtk.Button;
2 with Gtk.Box;
3 with Gtk.Label;
4 with Gtk.Image;
5
6 package Gtk.Key_Button is
7
8   type Gtk_Key_Button_Record is
9     new Gtk.Button.Gtk_Button_Record with private;
10  -- See private section.
11  -- Gtk_Key_Button_Record stores the position of the Key_Button
12  -- in a keypad by holding Row and Column number information
13  -- as Positive. Use functions Get_Button_Row and Get_Button_Column
14  -- to retrieve that information.
15
16  type Gtk_Key_Button is access all Gtk_Key_Button_Record'Class;
17  -- Access to all Gtk.Key_Button_Record widgets.
18
19  procedure Gtk_New (New_Key_Button : out Gtk_Key_Button;
20                   Label           : in  UTF8_String := "";
21                   Image           : in  Gtk.Image.Gtk_Image := null;
22                   Row             : in  Positive := 1;
23                   Column         : in  Positive := 1);
24  -- Creates new instance of Gtk_Key_Button_Record and sends
25  -- it to Initialise. Delivers initialized instance.
26
```



```
27  procedure Initialize
28      (New_Key_Button   : in out Gtk_Key_Button;
29       Label            : in      UTF8_String := "";
30       Image           : in      Gtk.Image.Gtk_Image := null;
31       Row             : in      Positive := 1;
32       Column          : in      Positive := 1);
33  -- Initializes Gtk_Key_Button_Record setting internal parameters
34  -- 'Row' and 'Column'.  If Image /= null, sets Icon from Image.
35  -- Sets Label if sent one.  Sets Name of button according to 'Row'
36  -- and 'Column'.  Convention is that, if 'Row' = 1, and 'Column' = 1
37  -- then Name will be "1_1".  Stores button's position in keypad.
38
39  function Get_Button_Row (Button : access Gtk_Key_Button_Record'Class)
40      return Positive;
41  -- Fetches the value of Button.Row.
42
43  function Get_Button_Column (Button : access Gtk_Key_Button_Record'Class)
44      return Positive;
45  -- Fetches the value of Button.Column.
46
47  private
48  type Gtk_Key_Button_Record is
49      new Gtk.Button.Gtk_Button_Record with
50      record
51          Label   : Gtk.Label.Gtk_Label := null;
52          Image   : Gtk.Image.Gtk_Image := null;
53          Box     : Gtk.Box.Gtk_HBox := null;
54          Row     : Positive := 1;
55          Column  : Positive := 1;
56      end record;
57 end Gtk.Key_Button;
```



9.1.2 Key Button Body

```
1 package body Gtk.Key_Button is
2
3   -- Creates new instance of Gtk_Key_Button_Record and sends
4   -- it to Initialise. Delivers initialized instance.
5   procedure Gtk_New (New_Key_Button :    out Gtk_Key_Button;
6                     Label           : in   UTF8_String := "";
7                     Image            : in   Gtk.Image.Gtk_Image := null;
8                     Row              : in   Positive := 1;
9                     Column           : in   Positive := 1) is
10  begin -- Gtk_New
11    New_Key_Button := new Gtk_Key_Button_Record;
12    Initialize (New_Key_Button => New_Key_Button,
13              Label           => Label,
14              Image           => Image,
15              Row             => Row,
16              Column          => Column);
17  end Gtk_New;
18
19  -- Initializes Gtk_Key_Button_Record setting internal parameters
20  -- 'Row' and 'Column'. If Image /= null, sets Icon from Image.
21  -- Sets Label if sent one. Sets Name of button according to 'Row'
22  -- and 'Column'. Convention is that, if 'Row' = 1, and 'Column' = 1
23  -- then Name will be "1_1". Stores button's position in keypad.
24  procedure Initialize
25    (New_Key_Button : in out Gtk_Key_Button;
26     Label          : in   UTF8_String := "";
27     Image          : in   Gtk.Image.Gtk_Image := null;
28     Row            : in   Positive := 1;
29     Column         : in   Positive := 1)is
30
```



```
31      -- 'use' clause makes use of operator '/=' legal.
32      use Gtk.Image;
33
34      -- 'Row_Char' and 'Col_Char' are used in setting button Name.
35      Row_Char : Character := Character'Val (Row + 48);
36      Col_Char : Character := Character'Val (Column + 48);
37
38      begin -- Initialize
39
40          -- Get a new button.
41          Gtk.Button.Initialize (Button => New_Key_Button,
42                               Label  => "");
43
44          -- Give the button a name.
45          Set_Name (Widget => New_Key_Button,
46                  Name    => Row_Char & '_' & Col_Char);
47
48          -- Save the button's position in the keypad.
49          New_Key_Button.Row := Row;
50          New_Key_Button.Column := Column;
51
52          -- Get a new box in button.
53          Gtk.Box.Gtk_New_HBox (Box      => New_Key_Button.Box,
54                               Homogeneous => false,
55                               Spacing    => 0);
56
57          -- Set Icon if Image /= null.
58          if Image /= null then
59              -- Put image in box.
60              Gtk.Box.Pack_Start (In_Box => New_Key_Button.Box,
61                                 Child  => Image,
62                                 Expand  => TRUE,
63                                 Fill    => TRUE,
64                                 Padding => 4);
65          end if; -- Image /= null
66
67          -- Set Label if sent one.
68          if Label /= "" then
69              -- Create the Label for button.
70              Gtk.Label.Gtk_New (Label => New_Key_Button.Label,
```



```
71         Str => Label);
72
73     -- Put label in box.
74     Gtk.Box.Pack_Start (In_Box => New_Key_Button.Box,
75                       Child   => New_Key_Button.Label,
76                       Expand  => TRUE,
77                       Fill    => TRUE,
78                       Padding => 4);
79     end if; -- Label /= ""
80
81     -- Put the box in the button.
82     Add (Container => New_Key_Button,
83         Widget     => New_Key_Button.Box);
84 end Initialize;
85
86 -- Fetches the value of Button.Row.
87 function Get_Button_Row (Button : access Gtk_Key_Button_Record'Class)
88     return Positive is
89 begin -- Get_Button_Row
90     return Button.Row;
91 end Get_Button_Row;
92
93 -- Fetches the value of Button.Column.
94 function Get_Button_Column (Button : access Gtk_Key_Button_Record'Class)
95     return Positive is
96 begin -- Get_Button_Column
97     return Button.Column;
98 end Get_Button_Column;
99
100 end Gtk.Key_Button;
```



9.1.3 Key Button Test

```
1 with Gtk.Main;
2 with Gtk.Window;
3 with Gtk.Box;
4 with Gtk.Enums;
5 with Gtk.Image;
6 with Gtk.Key_Button;
7 procedure Key_Button_Test is
8   Key_Window   : Gtk.Window.Gtk_Window;
9   Window_HBox  : Gtk.Box.Gtk_HBox;
10  Image        : Gtk.Image.Gtk_Image;
11  Key_Button   : Gtk.Key_Button.Gtk_Key_Button;
12 begin -- Key_Button_Test
13   Gtk.Main.Set_Locale;
14   Gtk.Main.Init;
15
16   Gtk.Window.Gtk_New (Key_Window, Gtk.Enums.Window_Toplevel);
17   Gtk.Window.Set_Title (Window => Key_Window,
18                        Title   => "Key_Button Test");
19   Gtk.Image.Gtk_New (Image     => Image,
20                    Filename => "./img/Up.gif");
21   Gtk.Key_Button.Gtk_New (New_Key_Button => Key_Button,
22                          Label         => "Go Up",
23                          Image         => Image,
24                          Row           => 2,
25                          Column       => 4);
26   Gtk.Box.Gtk_New_HBox (Window_HBox, TRUE, 2);
27   Gtk.Box.Pack_Start (Window_HBox, Key_Button);
28   Gtk.Window.Add (Key_Window, Window_HBox);
29   Gtk.Window.Show_All (Widget => Key_Window);
30
31   Gtk.Main.Main;
32 end Key_Button_Test;
```



9.2 Keypad

9.2.1 Walker_Keypad Specification

```
1 with GLib;
2 with Gtk.Table;
3 with Gtk.Box;
4 with Gtk.Label;
5 with Gtk.Handlers;
6 with Gtk.Key_Button;
7
8 pragma Elaborate_All (Gtk.Handlers);
9
10 package Walker_Keypad_Pkg is
11
12     KEYPAD_ROWS_COUNT : constant := 2;
13     -- The number of rows on a keypad.
14
15     KEYPAD_COLUMNS_COUNT : constant := 2;
16     -- The number of columns on a keypad.
17
18     type Keys_Type is array
19         (1 .. KEYPAD_ROWS_COUNT, 1 .. KEYPAD_COLUMNS_COUNT)
20     of Gtk.Key_Button.Gtk_Key_Button;
21     --Intermediate, addressable storage of keys for the keypad.
22
23     type Walker_Keypad_Record is new Gtk.Table.Gtk_Table_Record with private;
24     -- The keypad which is a set of keys displayed in a Table.
25
26     type Walker_Keypad is access all Walker_Keypad_Record'Class;
27     -- Access to the keypad and its children.
28
```



```
9  procedure Walker_New
30     (Walker : out Walker_Keypad);
31  -- Creates and initializes the keypad.
32
33  procedure Initialize
34     (Walker : access Walker_Keypad_Record'Class);
35  -- Initializes the keypad.
36
37  package Key_Button_Handler_Pkg is
38     new Gtk.Handlers.User_Callback (Gtk.Key_Button.Gtk_Key_Button_Record,
39                                     Walker_Keypad);
40  -- This package can be accessed externally and used to connect
41  -- callback procedures to specific key_buttons in a keypad.
42
43  function Get_Walker_Keypad_Button
44     (Keypad      : Walker_Keypad;
45      Row, Column : Positive) return Gtk.Key_Button.Gtk_Key_Button;
46  -- Returns an individual button from Pad.
47 private
48  type Walker_Keypad_Record is new Gtk.Table.Gtk_Table_Record with
49     record
50         Pad          : Keys_Type := ((others => null), (others => null));
51     end record;
52 end Walker_Keypad_Pkg;
```



9.2.2 Walker Keypad Body

```
1 with Gtk.Box;
2 with Gtk.Image;
3 with Gtk.Enums;
4 with GLib;
5
6 with Ada.Strings.Unbounded;
7
8 package body Walker_Keypad_Pkg is
9
10  -- Type to keep track of the names of the image files.
11  type Button_Arrow_Data_Type is
12    array (1 .. KEYPAD_ROWS_COUNT, 1 .. KEYPAD_COLUMNS_COUNT)
13    of Ada.Strings.Unbounded.Unbounded_String;
14
15  -- Store the names and addresses of the image files.
16  Button_Arrows : constant Button_Arrow_Data_Type :=
17    ((Ada.Strings.Unbounded.To_Unbounded_String (Source => ".\img\Right.gif"),
18     Ada.Strings.Unbounded.To_Unbounded_String (Source => ".\img\Down.gif")),
19     (Ada.Strings.Unbounded.To_Unbounded_String (Source => ".\img\Up.gif"),
20     Ada.Strings.Unbounded.To_Unbounded_String (Source => ".\img\Left.gif")));
21
22  -- Store the text for the button labels.
23  Button_Label_Text : constant Button_Arrow_Data_Type :=
24    ((Ada.Strings.Unbounded.To_Unbounded_String (Source => "Go Right"),
25     Ada.Strings.Unbounded.To_Unbounded_String (Source => "Go Down")),
26     (Ada.Strings.Unbounded.To_Unbounded_String (Source => "Go Up"),
27     Ada.Strings.Unbounded.To_Unbounded_String (Source => "Go Left")));
28
```



```
29  -- Creates and initializes the keypad.
30  procedure Walker_New
31  (Walker : out Walker_Keypad) is
32  begin -- Walker_New
33      Walker := new Walker_Keypad_Record;
34      Initialize (Walker => Walker);
35  end Walker_New;
36
37  -- Initializes the keypad.
38  procedure Initialize (Walker : access Walker_Keypad_Record'Class) is
39
40      -- 'use' clause lets the operator 'or' be used for
41      -- Gtk.Enums in call to Attach.
42      use Gtk.Enums;
43
44
45      -- Image to place in button.
46      Image : Gtk.Image.Gtk_Image;
47
48  begin -- Initialize
49
50      -- Initialize a Table to hold the Buttons.
51      -- No need to coerce the constants into GLib.Gint
52      -- because they are just Named Numbers.
53      Gtk.Table.Initialize (Widget      => Walker,
54                          Rows         => KEYPAD_ROWS_COUNT,
55                          Columns      => KEYPAD_COLUMNS_COUNT,
56                          Homogeneous => TRUE);
57
```



```
58     Rows_Loop :
59     for Row in 1 .. KEYPAD_ROWS_COUNT loop
60         Columns_Loop :
61         for Column in 1 .. KEYPAD_COLUMNS_COUNT loop
62
63             -- Create the new arrow image.
64             -- Button_Arrows gives the appropriate address for the image.
65             Gtk.Image.Gtk_New
66             (Image      => Image,
67              Filename => Ada.Strings.Unbounded.To_String
68                (Button_Arrows (Row, Column)));
69
70             -- Create a new button in the Pad (the array).
71             -- It will already have a name, its coordinates in the
72             -- keypad, a label and an icon by initialisation.
73             -- Button_Label_Text gives the appropriate text for the label.
74             Gtk.Key_Button.Gtk_New
75             (New_Key_Button => Walker.Pad (Row, Column),
76              Label          => Ada.Strings.Unbounded
77                .To_String (Button_Label_Text (Row, Column)),
78              Image         => Image,
79              Row           => Row,
80              Column       => Column);
81
```



```
82         -- Attach the button to the Table.
83         -- Procedure inherited from Gtk.Table.
84         Attach (Table      => Walker,
85                Child       => Walker.Pad (Row, Column),
86                Left_Attach => GLib.GUint (Column - 1),
87                Right_Attach => GLib.GUint (Column),
88                Top_Attach  => GLib.GUint (Row - 1),
89                Bottom_Attach => GLib.GUint (Row),
90                Xoptions    => Expand or Fill,
91                Yoptions    => Expand or Fill,
92                Xpadding    => 20,
93                Ypadding    => 20);
94
95         end loop Columns_Loop;
96     end loop Rows_Loop;
97 end Initialize;
98
99 -- Returns an individual button from Pad.
100 function Get_Walker_Keypad_Button
101     (Keypad      : Walker_Keypad;
102      Row, Column : Positive) return Gtk.Key_Button.Gtk_Key_Button is
103 begin -- Get_Walker_Keypad_Button
104     return Keypad.Pad (Row, Column);
105 end Get_Walker_Keypad_Button;
106 end Walker_Keypad_Pkg;
```



9.2.3 Walker Keypad Test

```
1 with Gtk.Window;
2 with Gtk.Enums;
3 with Gtk.Box;
4 with Gtk.Main;
5 with Walker_Keypad_Pkg;
6 procedure Walker_Keypad_Test is
7   Test_Window : Gtk.Window.Gtk_Window;
8   Keypad_Box   : Gtk.Box.Gtk_HBox;
9   Keypad       : Walker_Keypad_Pkg.Walker_Keypad;
10 begin -- Walker_Keypad_Test
11   Gtk.Main.Set_Locale;
12   Gtk.Main.Init;
13
14   Gtk.Window.Gtk_New (Window => Test_Window,
15                       The_Type => Gtk.Enums.Window_Toplevel);
16   Gtk.Box.Gtk_New_HBox (Box      => Keypad_Box,
17                        Homogeneous => TRUE,
18                        Spacing   => 0);
19   Walker_Keypad_Pkg.Walker_New (Walker => Keypad);
20   Gtk.Box.Pack_Start (In_Box => Keypad_Box,
21                     Child   => Keypad,
22                     Expand  => TRUE,
23                     Fill    => TRUE,
24                     Padding => 2);
25   Gtk.Window.Add (Container => Test_Window,
26                  Widget    => Keypad_Box);
27
28   Gtk.Window.Show_All (Widget => Test_Window);
29
30   Gtk.Main.Main;
31 end Walker_Keypad_Test;
```

9.3 Walker

9.3.1 Walker_Pkg Specification



```
1 with Gtk.Window;
2 with Gtk.Box;
3 with Walker_Keypad_Pkg;
4 package Walker_Pkg is
5
6   type Walker_Record is
7     new Gtk.Window.Gtk_Window_Record with
8     record
9       Keypad_Box : Gtk.Box.Gtk_Hbox := null;
10      Keypad      : Walker_Keypad_Pkg.Walker_Keypad := null;
11    end record;
12   -- Define a window which will have a Walker Keypad in it.
13
14   type Walker_Access is access all Walker_Record'Class;
15   -- Define access to the window with the keypad.
16
17   procedure Gtk_New (Widget : out Walker_Access);
18   -- Create and initialize a new Walker Window.
19
20   procedure Initialize (Widget : access Walker_Record'Class);
21   -- Initialize a Walker Window.
22 end Walker_Pkg;
```



9.3.2 Walker Pkg Body

```
1 with Gtk.Enums;
2 with Gtk.Widget;
3
4 with Walker_Pkg.Callbacks;
5 with Walker_Pkg_Callback_Pkg;
6 with Walker_Keypad_Pkg.Local_Callbacks;
7
8 package body Walker_Pkg is
9
10  procedure Gtk_New (Widget :    out Walker_Access) is
11  begin -- Gtk_New
12      Widget := new Walker_Record;
13      Initialize (Widget => Widget);
14  end Gtk_New;
15
16  procedure Initialize (Widget : access Walker_Record'Class) is
17      Icon : Boolean;
18  begin -- Initialize
19
20      -- Create the window.
21      Gtk.Window.Initialize (Window  => Widget,
22                            The_Type => Gtk.Enums.Window_Toplevel);
23
24      -- Give the window a Title.
25      -- Procedure inherited from Gtk.Window;
26      Set_Title (Window => Widget,
27                Title  => "Walker");
28
29      -- Give the window an Icon.
30      -- Function inherited from Gtk.Window;
31      Icon := Set_Icon_From_File (Window  => Widget,
32                                 Filename => "./img/Walker.gif");
33
34      -- Connect the window destructor.
35      Walker_Pkg_Callback_Pkg.Window_Callback_Pkg.Connect
36      (Widget => Widget,
37       Name  => "destroy",
```



```
38     Marsh => Walker_Pkg_Callback_Pkg.Window_Callback_Pkg
39     .To_Marshaller
40     (Walker_Pkg.Callbacks.On_X_Button_Destroy'Access));
41
42 -- Put the window at the start position.
43 Gtk.Widget.Set_UPosition
44 (Widget => Get_Toplevel (Widget => Widget),
45  X      => 100,
46  Y      => 100);
47
48 -- Create keypad box
49 Gtk.Box.Gtk_New_Hbox (Box          => Widget.Keypad_Box,
50                      Homogeneous => TRUE,
51                      Spacing      => 0);
52
53 -- Put box in window.
54 Add (Container => Widget,
55     Widget     => Widget.Keypad_Box);
56
57 -- Create the keypad.
58 Walker_Keypad_Pkg.Walker_New (Walker => Widget.Keypad);
59
60 -- Connect the callbacks to the keys in the keypad.
61 Row_Loop :
62 for Row in
63   1 .. Walker_Keypad_Pkg.KEYPAD_ROWS_COUNT
64 loop
65   Column_Loop :
66   for Column in
67     1 .. Walker_Keypad_Pkg.KEYPAD_COLUMNS_COUNT
68 loop
69   Walker_Keypad_Pkg.Key_Button_Handler_Pkg.Connect
70   (Widget => Walker_Keypad_Pkg.Get_Walker_Keypad_Button
71    (Keypad => Widget.Keypad,
72     Row   => Row,
73     Column => Column),
74    Name  => "clicked",
75    Marsh =>
76     Walker_Keypad_Pkg.Key_Button_Handler_Pkg
77     .To_Marshaller
```



```
78         (Walker_Keypad_Pkg.Local_Callbacks
79          .On_Walker_Keypad_Button_Click'Access),
80         User_Data => Widget.Keypad);
81     end loop Column_Loop;
82 end loop Row_Loop;
83
84 -- Install keypad in box in window.
85 Gtk.Box.Pack_Start (In_Box => Widget.Keypad_Box,
86                   Child  => Widget.Keypad,
87                   Expand => TRUE,
88                   Fill   => TRUE,
89                   Padding => 0);
90 end Initialize;
91 end Walker_Pkg;
```



9.3.3 Walker_Pkg_Callback_Pkg Specification

```
1 with Gtk.Handlers;
2 with Gtk.Window;
3
4 pragma Elaborate_All (Gtk.Handlers);
5
6 package Walker_Pkg_Callback_Pkg is
7     package Window_Callback_Pkg is new
8         Gtk.Handlers.Callback (Gtk.Window.Gtk_Window_Record);
9     -- This instantiation creates a new package to handle
10    -- Gtk_Window_Record.
11 end Walker_Pkg_Callback_Pkg;
```

9.3.4 Walker_Pkg.Callbacks Specification

```
1 with Gtk.Window;
2 package Walker_Pkg.Callbacks is
3     procedure On_X_Button_Destroy
4         (Object : access Gtk.Window.Gtk_Window_Record'Class);
5     -- On_X_Button_Destroy is a callback to destroy the main window and end
6     -- the program run when the [X] button at top right of the window is clicked.
7 end Walker_Pkg.Callbacks;
```

9.3.5 Walker_Pkg.Callbacks Body

```
1 with Gtk.Main;
2 package body Walker_Pkg.Callbacks is
3     -- On_X_Button_Destroy is a callback to destroy the main window and end
4     -- the program run when the [X] button at top right of the window is clicked.
5     procedure On_X_Button_Destroy
6         (Object : access Gtk.Window.Gtk_Window_Record'Class) is
7     begin -- On_X_Button_Destroy
8         Gtk.Main.Main_Quit;
9     end On_X_Button_Destroy;
10 end Walker_Pkg.Callbacks;
```



9.3.6 Walker Keypad_Pkg.Local_Callbacks Specification

```
1 with Gtk.Window;
2 with Gtk.Key_Button;
3 package Walker_Keypad_Pkg.Local_Callbacks is
4   procedure On_Walker_Keypad_Button_Click
5     (Button : access Gtk.Key_Button.Gtk_Key_Button_Record'Class;
6      Keypad : Walker_Keypad);
7   -- On_Phone_Keypad_Button_Click reads the Row and Column of the
8   -- Keypad_Button and takes appropriate action on that basis.
9 end Walker_Keypad_Pkg.Local_Callbacks;
```



9.3.7 Walker Keypad Pkg.Local Callbacks Body

```
1 with Gtk.Main;
2 with Gtk.Widget;
3 with GLib;
4 with Gtk.Key_Button;
5 with Walker_Pkg_Operations;
6 with Walker_Pkg;
7 package body Walker_Keypad_Pkg.Local_Callbacks is
8   -- On_Phone_Keypad_Button_Click reads the Row and Column of the
9   -- Keypad_Button and takes appropriate action on that basis.
10  procedure On_Walker_Keypad_Button_Click
11    (Button : access Gtk.Key_Button.Gtk_Key_Button_Record'Class;
12     Keypad : Walker_Keypad) is
13
14     -- Find the X coordinate of the top-left-hand corner of the
15     -- toplevel widget (the window).
16     X_Pos : GLib.Gint := Gtk.Widget.Get_Allocation_X
17       (Widget => Get_Toplevel (Widget => Keypad));
18
19     -- Find the Y coordinate of the top-left-hand corner of the
20     -- toplevel widget (the window).
21     Y_Pos : GLib.Gint := Gtk.Widget.Get_Allocation_Y
22       (Widget => Get_Toplevel (Widget => Keypad));
23
24     -- Identify the toplevel widget (the window).
25     Window : Gtk.Widget.Gtk_Widget :=
26       Gtk.Key_Button.Get_Toplevel (Widget => Button);
27
```



```
28 begin -- On_Walker_Keypad_Button_Click
29
30     -- Call the appropriate operation via the array
31     -- of Operation_Access_Type.
32     Walker_Pkg_Operations.Operation_Access_Array
33     (Gtk.Key_Button.Get_Button_Row (Button), -- Row coordinate in array
34      Gtk.Key_Button.Get_Button_Column (Button) -- Column coordinate in array
35     )
36     (Window, X_Pos, Y_Pos);
37
38 end On_Walker_Keypad_Button_Click;
39 end Walker_Keypad_Pkg.Local_Callbacks;
```

9.3.8 Walker_Keypad_Operations Specification

```
1 with GLib;
2 with Gtk.Widget;
3 with Walker_Pkg;
4 with Walker_Keypad_Pkg;
5 package Walker_Pkg_Operations is
6
7     type Operation_Access_Type is access
8         procedure (Window : Gtk.Widget.Gtk_Widget;
9                  X, Y : in GLib.Gint);
10    -- Define an access to the procedures required in response to
11    -- the clicking of a button on the keypad.
12
13    type Operation_Access_Array_Type is
14        array (1 .. Walker_Keypad_Pkg.KEYPAD_ROWS_COUNT,
15              1 .. Walker_Keypad_Pkg.KEYPAD_COLUMNS_COUNT)
16    of Operation_Access_Type;
17    -- Define an array of procedure access to match the keypad.
18
19    procedure Button_1_1 (Window : Gtk.Widget.Gtk_Widget;
20                       X, Y : in GLib.Gint);
21    -- Define operation in response to click on 'Go Right' button.
22
23    procedure Button_1_2 (Window : Gtk.Widget.Gtk_Widget;
```



```
24         X, Y : in GLib.Gint);
25 -- Define operation in response to click on 'Go Down' button.
26
27 procedure Button_2_1 (Window : Gtk.Widget.Gtk_Widget;
28         X, Y : in GLib.Gint);
29 -- Define operation in response to click on 'Go Up' button.
30
31 procedure Button_2_2 (Window : Gtk.Widget.Gtk_Widget;
32         X, Y : in GLib.Gint);
33 -- Define operation in response to click on 'Go Left' button.
34
35 Operation_Access_Array : Operation_Access_Array_Type :=
36     ((Button_1_1'Access, Button_1_2'Access),
37      (Button_2_1'Access, Button_2_2'Access));
38 -- Assign appropriate procedure access to each cell of the array.
39 end Walker_Pkg_Operations;
```



9.3.9 Walker Keypad Operations Body

```
1 with Gtk.Widget;
2 package body Walker_Pkg_Operations is
3
4   -- Operation in response to 'Go Right' button.
5   procedure Button_1_1 (Window : Gtk.Widget.Gtk_Widget;
6                        X, Y    : in   GLib.Gint) is
7     -- Set for right horizontal position.
8     X_Pos : GLib.Gint := 400;
9
10    -- set for existing vertical position.
11    Y_Pos : GLib.Gint := Y;
12  begin -- Button_1_1
13    -- Move the window.
14    Gtk.Widget.Set_UPosition
15      (Widget => Window,
16       X      => GLib.Gint (X_Pos),
17       Y      => GLib.Gint (Y_Pos));
18  end Button_1_1;
19
20
21  -- Operation in response to 'Go Down' button.
22  procedure Button_1_2 (Window : Gtk.Widget.Gtk_Widget;
23                      X, Y    : in   GLib.Gint) is
24    -- Set for existing horizontal position.
25    X_Pos : GLib.Gint := X;
26
27    -- Set for lower vertical position.
28    Y_Pos : GLib.Gint := 400;
```



```
29  begin -- Button_1_2
30      -- Move the window.
31      Gtk.Widget.Set_UPosition
32          (Widget => Window,
33           X      => GLib.Gint (X_Pos),
34           Y      => GLib.Gint (Y_Pos));
35  end Button_1_2;
36
37
38  -- Operation in response to 'Go Up' button.
39  procedure Button_2_1 (Window : Gtk.Widget.Gtk_Widget;
40                      X, Y    : in  GLib.Gint) is
41      -- Set for existing horizontal position.
42      X_Pos : GLib.Gint := X;
43
44      -- Set for upper vertical position.
45      Y_Pos : GLib.Gint := 100;
46  begin -- Button_2_1
47      -- Move the window.
48      Gtk.Widget.Set_UPosition
49          (Widget => Window,
50           X      => GLib.Gint (X_Pos),
51           Y      => GLib.Gint (Y_Pos));
52  end Button_2_1;
53
54
55  -- Operation in response to 'Go Left' button.
56  procedure Button_2_2 (Window : Gtk.Widget.Gtk_Widget;
57                      X, Y    : in  GLib.Gint) is
58      -- Set for left horizontal position.
59      X_Pos : GLib.Gint := 100;
60
```



```
61      -- Set for existing vertical position.
62      Y_Pos : GLib.Gint := Y;
63  begin -- Button_2_2
64      -- Move the window.
65      Gtk.Widget.Set_UPosition
66      (Widget => Window,
67       X      => GLib.Gint (X_Pos),
68       Y      => GLib.Gint (Y_Pos));
69  end Button_2_2;
70
71 end Walker_Pkg_Operations;
```

9.3.10 Walker — Mainline

```
1  with Gtk.Main;
2  with Walker_Pkg;
3  procedure Walker is
4      Walker : Walker_Pkg.Walker_Access;
5  begin -- Walker
6      Gtk.Main.Set_Locale;
7      Gtk.Main.Init;
8
9      Walker_Pkg.Gtk_New (Widget => Walker);
10     Walker_Pkg.Show_All (Widget => Walker);
11
12     Gtk.Main.Main;
13 end Walker;
```



10 Access Types as Parameters

The question to CompLangAda was:

Given:

```
type My_Type is ...;
and
type My_Access_Type is access all My_Type'Class;
```

what is the practical difference between:

```
function My_Function (Thing : access My_Type'Class)
return Positive;
and
function My_Function (Thing : My_Access_Type)
return Positive;
```

Adam Benesch replied:

The accessibility level rules say that a value of type My_Access_Type cannot point to an object that might go away before My_Access_Type does. (Unless you use Unchecked_Access to create the access value.) This is to prevent dangling references. The rules apply to parameters also, so if you call the second My_Function, you can't give it the 'Access of an object that is deeper than My_Access_Type.

If My_Access_Type is declared at library level (i.e. in a library package, not inside a procedure or function), then you can't say

```
procedure Some_Procedure is
X : aliased My_Type; --or some type derived from My_Type
begin
My_Function (X'access); -- i.e. the second My_Function
```

But this restriction doesn't apply to the first My_Function. Since that My_Function has an access parameter, you can pass the 'Access of any object of the right type, no matter how deeply nested inside procedures the object is.

Dmitry A. Kazakov added:

> *Why then does GtkAda consistently use the first form,*

```
function My_Function (Thing : access Gtk_XXX_Record'Class)
return Positive;
```

I think it should better be

```
function My_Function (Thing : Gtk_XXX) return Positive;
```

since GtkAda has Gtk_XXX declared as access to every Gtk_XXX_Record'Class.

... this is likely a design bug. Gtk_Object_Record should have been limited controlled with Initialize inherited from Ada.Finalization.Limited_Controlled.

Dmitry Kazakov added:

```
function My_Function (Thing : access Gtk_XXX_Record'Class)
return Positive;
```



I think it should better be

```
function My_Function (Thing : Gtk_XXX) return Positive;
```

since GtkAda has Gtk_XXX declared as access to every Gtk_XXX_Record'Class.

```
> It actually uses the form (this for Gtk.Button(Gtk_Button):  
> procedure Initialize  
>   (Button : access Gtk_Button_Record'Class;  
>   Label   : UTF8_String);
```

Well, this is likely a design bug. Gtk_Object_Record should have been limited controlled with Initialize inherited from Ada.Finalization.Limited_Controlled.

Adam added:

*Well, for a *function*, using an access parameter can help get around the rule that you can't have an IN OUT parameter to a function (a rule that I think is going away in the next revision of the language). By making it an anonymous access type, rather than a named access type, that lets you pass local variables that the function can modify.*

For a procedure, there's less reason to do so. I don't know anything about GTK, so I don't know why this wouldn't have worked, if all Initialize is doing is to set up some fields in Button:

```
procedure Initialize  
  (Button : in out Gtk_Button_Record'Class;  
   Label   : UTF8_String);
```

If, on the other hand, Initialize needs Button as an access so that it can store it in a data structure, then it probably would have been best to make it a named access type, to ensure that dangling references aren't stored in that data structure.

*If Initialize is an imported routine from a C library---well, in that case, I guess you do whatever works. But I'd still think that using "in out" would have the same effect---the address of the record is passed. (In fact, for an anonymous access parameter, the code normally CANNOT pass *just* an address; there has to be additional information in case the procedure needs to do an accessibility level check. But for a procedure imported from C, the parameter passing mechanisms may be different.)*

Randy is right, and I thought about saying something along those lines in my first response but decided (maybe unwisely) not to complicate things. You don't need to pass an anonymous access parameter if all you're going to do is modify the accessed object---IN OUT will do. And if you're going to store the access parameter in a global structure, you'll get a runtime error if you try to store a dangling reference, so it's best to use a global access type so that accessibility level errors are caught at compile time. There may be some esoteric situations, other than dispatching, where passing an anonymous access parameter may be useful: perhaps the procedure can determine at runtime whether to store an access value or make a copy of the object depending on the accessibility level, or perhaps the procedure wants the ability to accept NULL to indicate that there is no object to modify, or perhaps a procedure takes two access parameters and sets up the two accessed objects to point to each other. But those all seem very unusual, and may be due to an inferior design anyway.

Stephen Leake, perhaps, explained it best:

DOCUMENT CURRENT ONLY WHEN PRINTED

e:\archive\labbook\walker.docx
Printed on Tuesday, 1 September 2009



Most GTK subprograms do need access values; the same is true of any system that deals with derived types at multiple levels. The only library-level object that can hold any type in the hierarchy is a class-wide pointer. Lists of objects must hold pointers, etc.

So one reason to use 'access' instead of 'in out' is simply to avoid the user having to type '.all' everywhere.

A reason to use 'access Record_Type' instead of 'in Access_Type' is to avoid explicit type conversions. Given:

```
package Widget is
  type Gtk_Widget_Record is ...;
  type Gtk_Widget is access all Gtk_Widget_Record'class;
  procedure Show (Widget : in Gtk_Widget);
end Widget;
package Window is
  type Gtk_Window_Record is new Gtk_Widget_Record with ...;
  type Gtk_Window is access all Gtk_Window_Record'class;
end Window;
  Window : Gtk_Window := ...;
```

then this does not work:

```
Widget.Show (Window);
```

but this does:

```
Widget.Show (Gtk_Widget (Window));
```

This is just annoying!

However, if Show is declared:

```
procedure Show (Widget : access constant Gtk_Widget_Record'class);
```

Then Show matches any access type in the class hierarchy;

```
Widget.Show (Window);
```

works.



11 Reference List

- Barnes, J. (2006). *Programming in Ada 2005*. Harlow, UK: Pearson Education Ltd.
- Dale, N., Weems, C., & McCormick, J. (2000). *Programming and Problem Solving with Ada95 (second edition)*. Sudbury, MA (USA): Jones and Bartlett.
- Software Productivity Consortium (1995). *Ada 95 Quality and Style: Guidelines for Professional Programmers*. Department of Defense Ada Joint Program Office.

